********************************************************************************
* Purpose: this file creates all robustness output 
********************************************************************************
* Robustness: Sample restriction
* Alternative treatments: share
* With covariates. both individual and group-level characteristics
* Include small groups
* Vary screening age to be included in treatment definition
********************************************************************************
do "D:\Data\workdata\708296\Project - Mother groups and mental health\Do\global.do" 


********************************************************************************
* Figure C10 Robustness: Percentage coverage of Mother group with EPDS screening
********************************************************************************

clear
cap program drop specchart
program specchart
syntax varlist, [replace] spec(string)
tempfile temp
save "`temp'", replace
if  "`replace'"!=""{
                clear
				gen beta=.
				gen se=.
				gen spec_id=.
				gen u95=.
				gen u90=.
				gen l95=.
				gen l90=.
				save "estimates.dta",replace
}
else{
use "estimates.dta",clear
}
local obs=_N+1
set obs `obs'
replace spec_id =`obs' if _n == `obs'
replace beta = _b[`varlist'] if spec_id ==`obs'				
replace se = _se[`varlist'] if spec_id ==`obs'	
replace u95 = beta+1.96*se if spec_id ==`obs'	
replace u90 = beta+1.645*se if spec_id ==`obs'	
replace l95 = beta-1.96*se if spec_id ==`obs'	
replace l90 = beta-1.645*se if spec_id ==`obs'	
foreach s in `spec'{
cap gen `s'=1 if spec_id ==`obs'
cap replace `s'=1 if spec_id == `obs'
}
save "estimates.dta", replace
use `temp', clear
end
********************************************************************************
* Add outcomes

use "$work\final_data", clear
areg anyw24 peer_treat if tested_avg >= 1, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(main anyw24 sample10) replace

forvalues i = 0(1)9 {
areg anyw24 peer_treat if tested_avg >= `i'/10, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(anyw24 sample`i')
}
use "$work\final_data", clear
areg consw24 peer_treat if tested_avg >= 1, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(main consw24 sample10) 

forvalues i = 0(1)9 {
areg consw24 peer_treat if tested_avg >= `i'/10, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(consw24 sample`i')
}
use "$work\final_data", clear
areg psykw24 peer_treat if tested_avg >= 1, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(main psykw24 sample10) 

forvalues i = 0(1)9 {
areg psykw24 peer_treat if tested_avg >= `i'/10, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(psykw24 sample`i')
}
use "$work\final_data", clear
areg adm24 peer_treat if tested_avg >= 1, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(main adm24 sample10)

forvalues i = 0(1)9 {
areg adm24 peer_treat if tested_avg >= `i'/10, absorb(time fe mfr_first) vce(cluster group)
specchart peer_treat , spec(adm24 sample`i')
}


use "estimates.dta", clear
duplicates drop
gen xx = 0 if anyw24 == 1
replace xx = 1 if consw24 == 1
replace xx = 2 if psykw24 == 1 
replace xx = 3 if adm24 == 1
sort xx beta
	  
gen rank = _n
local scoff = " "
local scon= " "
local ind= -0.004
*foreach var in quad linear rdd ols poly {
foreach	 var in anyw24 consw24 psykw24 adm24 {
cap gen i_`var' = `ind'
local ind = `ind'-0.002
local scoff="`scoff' (scatter i_`var' rank,msize(small) mcolor(gs14))"
local scon="`scon' (scatter i_`var' rank if `var'==1,msize(small) mcolor(black))"
}
local ind=`ind'-0.002
forvalues i = 0(1)10  {
*foreach i in 36 40 44 48 52 56 60 opt {	
cap gen i_sample`i'=`ind'
local ind=`ind'-0.002
local scoff"`scoff' (scatter i_sample`i' rank,msize(small) mcolor(gs14))"
local scon"`scon' (scatter i_sample`i' rank if sample`i'==1,msize(small) mcolor(black))"
} 
tw (scatter beta rank if main==1, mcolor(blue) msymbol(D) msize(small)) ///
   (rbar u95 l95 rank, fcolor(gs12) lcolor(gs12) lwidth(none)) ///
   (scatter beta rank, mcolor(black) msymbol(D) msize(small)) ///
   `scoff' `scon' ///
   (scatter beta rank if main==1, mcolor(blue) msymbol(D) msize(small)) ///
   (scatter i_sample10 rank if main==1, msize(vsmall) mcolor(blue))  , graphregion(color(white) lcolor(white)) plotregion(color(white) lcolor(white)) ///
   xtitle(" ") ytitle(" ") yscale(noline) xscale(noline) ylab(-0.01(0.01)0.03,noticks nogrid angle(horizontal)) xlab("",noticks) legend(order(1 "Main spec." 3 "Point estimate" 2 "95% CI") region(lcolor(white)) pos(12) ring(1) rows(1) size(vsmall) symysize(small) symxsize(small)) ///
   graphregion(color(white) lcolor(white)) plotregion(color(white) lcolor(white)) bgcolor(white) yline(0)
   
   gr_edit .yaxis1.add_ticks -0.002 `"Outcome (at age 2) "', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.004 `"Mental health contact"', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.006 `"GP mental health consultation"', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.008 `"Psychologist/psychiatrist"', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.01 `"Mental health hospitalization"', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))

  gr_edit .yaxis1.add_ticks -0.012 `"Percent tested in groups "', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.014 `"No restrictions"', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.016 `"10 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.018 `"20 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.02 `"30 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.022 `"40 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.024 `"50 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.026 `"60 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.028 `"70 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.03 `"80 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.032 `"90 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
  gr_edit .yaxis1.add_ticks -0.034 `"100 pct."', custom tickset(major) editstyle(tickstyle(textstyle(size(vsmall))))
graph export "$out\Speccurveall.png", replace



********************************************************************************
* Table C21: Robustness: Omit groups with late screeening of the depressed peer
********************************************************************************
*Drop groups with depressed peer tested later than three months after birth

use "$work\final_data", clear
replace score_e1 = . if age_test1 > 90 & high == 1
drop high_mean high peer_treat tested_avg tested

gen high = 0
replace high = 1 if score_e1 > 10
replace high = 0 if score_e1 == .

gen tested = 0
replace tested = 1 if score_e1 != .

rangestat high, int(group 0 0) excludeself
egen tested_avg = mean(tested), by(group)

gen peer_treat = 0
replace peer_treat = 1 if high_mean > 0

keep if tested_avg >= 1 // all tested
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat , absorb(time fe mfr_first) vce(cluster group)
estimates store p`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab p* using "$out\mentalhealth_regdate.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace

********************************************************************************
* Table C20: Robustness: Municipality or nurse fixed effects
********************************************************************************

* Alternative FE -  among all tested  
use "$work\final_data1", clear
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat , absorb(time mfr_muni mfr_first) vce(cluster group)
estimates store p`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab p* using "$out\mentalhealth_muniFE.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace

use "$work\final_data1", clear
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat , absorb(time nurse mfr_first) vce(cluster group)
estimates store p`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab p* using "$out\mentalhealth_nurseFE.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace


// Alternative treatments 

********************************************************************************
* Table C15: Robustness: Average screening score of peers as treatment variable
********************************************************************************
use "$work\final_data1", clear
replace peer_treat = score_e1_mean
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat , absorb(time fe mfr_first) vce(cluster group)
estimates store p`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab p* using "$out\mentalhealth_altppeer.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace

********************************************************************************
* Table C16: Robustness: Share of peers with high screening score as treatment variable
********************************************************************************
replace peer_treat = high_mean
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat , absorb(time fe mfr_first) vce(cluster group)
estimates store p`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab p* using "$out\mentalhealth_altppeer2.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace


********************************************************************************
* Table C17: Robustness: Omit contacts in the two first months after birth
********************************************************************************
use "$work\final_data1", clear
eststo clear
foreach var in any2_robust gpm_cons_robust gpm_psyk_robust postbirth2y_robust {
eststo: areg `var' peer_treat , absorb(time fe mfr_first) vce(cluster group)
estimates store q`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab q* using "$out\mentalhealth_dropcontacts1.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace

********************************************************************************
* Table C18: Robustness: Controls for predetermined birth and parental characteristics
********************************************************************************

use "$work\final_data1", clear
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat $covariates , absorb(time fe mfr_first) vce(cluster group)
estimates store z`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab z* using "$out\mentalhealth_covar.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace


********************************************************************************
* Table C19: Robustness: Controls for average peer characteristics
********************************************************************************

use "$work\final_data1", clear
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat mfr_cs_mean mfr_paritet_mean mfr_female_mean mfr_homebirth_mean mfr_first_mean mfr_lowbw_mean mfr_preterm_mean inc_m_mean inc_f_mean grund_m_mean grund_f_mean higher_m_mean higher_f_mean uni_m_mean uni_f_mean dk_m_mean dk_f_mean married_mean cohab_mean alder_m_mean alder_f_mean , absorb(time fe mfr_first) vce(cluster group)
estimates store w`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab w* using "$out\mentalhealth_groupchar.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace

********************************************************************************
* Table C22: Robustness: Omit outcome data of individuals with a screening score above 10
********************************************************************************

use "$work\final_data1", clear
drop if high == 1
eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat , absorb(time fe mfr_first) vce(cluster group)
estimates store aa`var'
summarize `var' if peer_treat == 0, meanonly
estadd scalar means = r(mean)
}
esttab aa* using "$out\mentalhealth_drophigh.tex", fragment b(3) se(3) keep(peer_treat) stats(means N, fmt(%13.2fc %13.0fc) label("Control group mean" "Obs.")) se star(* 0.10 ** 0.05 *** 0.01) alignment(c c) label nolegend nonote replace

********************************************************************************
* Figure C11: Timing of EPDS screening
********************************************************************************

* Curve with varying restrictions on EPDS screening date. From week 5 to week 20
use "$work\final_data1", clear

tab age_test1
forvalues i = 5/20 {
	preserve
drop high_mean
replace high = 0 if age_test1 > `i'*7
rangestat high, int(group 0 0) excludeself
gen peer_treat`i' = 0
replace peer_treat`i' = 1 if high_mean > 0

eststo clear
foreach var in anyw24 consw24 psykw24 adm24 {
eststo: areg `var' peer_treat`i', absorb(time fe mfr_first) vce(cluster group)
parmest, format(estimate min95 max95 %8.3f p %8.1e) saving("$work\robust_`var'_epdsweek`i'", replace)
}
restore
}

use "$work\robust_consw24_epdsweek5",clear
gen w = 5
forvalues i = 6/20 {
append using "$work\robust_consw24_epdsweek`i'"
replace w = `i' if w == .
}
drop if parm == "_cons"
replace t = round(t,0.001)
replace stderr = round(stderr,0.001)
twoway (scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(estimate) mlabp(12) mlabs(vsmall) mlabg(15pt) mlabc(black)) /// 
(scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(stderr) mlabp(12) mlabs(vsmall) mlabg(6pt) mlabc(black)) ///
(line estimate w, fcolor(black%10) alwidth(none) lcolor(red) lwidth(thick)) ///
(line max95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)) ///
(line min95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)), ylab(,labcolor(black) glcolor(gs10%33) angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) xlab(5 "5" 10 "10" 15 "15" 20 "20", labcolor(black) glcolor(gs10%33) grid angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) graphregion(color(white)) bgcolor(white) ysize(7) xsize(10) yscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) xscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) graphregion(lcolor(white) fcolor(white)) xtitle(,color(black)) ytitle(,color(black)) legend(off) xtitle("Exclude screening scores above X weeks") ytitle("Estimate") ylab(-0.04(0.02)0.06, tlcolor(gs10) labgap(6-pt) format(%4.2f)) yline(0, lcolor(gs10) lwidth(medthick)) yline(0.015, lcolor(blue) lwidth(medthick)) 
graph export "$out\cons_epdsdate.png", replace

use "$work\robust_psykw24_epdsweek5",clear
gen w = 5
forvalues i = 6/20 {
append using "$work\robust_psykw24_epdsweek`i'"
replace w = `i' if w == .
}
drop if parm == "_cons"
replace t = round(t,0.001)
replace stderr = round(stderr,0.001)
twoway (scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(estimate) mlabp(12) mlabs(vsmall) mlabg(15pt) mlabc(black)) /// 
(scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(stderr) mlabp(12) mlabs(vsmall) mlabg(6pt) mlabc(black)) ///
(line estimate w, fcolor(black%10) alwidth(none) lcolor(red) lwidth(thick)) ///
(line max95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)) ///
(line min95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)), ylab(,labcolor(black) glcolor(gs10%33) angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) xlab(5 "5" 10 "10" 15 "15" 20 "20", labcolor(black) glcolor(gs10%33) grid angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) graphregion(color(white)) bgcolor(white) ysize(7) xsize(10) yscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) xscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) graphregion(lcolor(white) fcolor(white)) xtitle(,color(black)) ytitle(,color(black)) legend(off) xtitle("Exclude screening scores above X weeks") ytitle("Estimate") ylab(-0.04(0.02)0.06, tlcolor(gs10) labgap(6-pt) format(%4.2f)) yline(0, lcolor(gs10) lwidth(medthick)) yline(0.007, lcolor(blue) lwidth(medthick)) 
graph export "$out\psyk_epdsdate.png", replace

use "$work\robust_anyw24_epdsweek5",clear
gen w = 5
forvalues i = 6/20 {
append using "$work\robust_anyw24_epdsweek`i'"
replace w = `i' if w == .
}
drop if parm == "_cons"
replace t = round(t,0.001)
replace stderr = round(stderr,0.001)
twoway (scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(estimate) mlabp(12) mlabs(vsmall) mlabg(15pt) mlabc(black)) /// 
(scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(stderr) mlabp(12) mlabs(vsmall) mlabg(6pt) mlabc(black)) ///
(line estimate w, fcolor(black%10) alwidth(none) lcolor(red) lwidth(thick)) ///
(line max95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)) ///
(line min95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)), ylab(,labcolor(black) glcolor(gs10%33) angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) xlab(5 "5" 10 "10" 15 "15" 20 "20", labcolor(black) glcolor(gs10%33) grid angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) graphregion(color(white)) bgcolor(white) ysize(7) xsize(10) yscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) xscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) graphregion(lcolor(white) fcolor(white)) xtitle(,color(black)) ytitle(,color(black)) legend(off) xtitle("Exclude screening scores above X weeks") ytitle("Estimate") ylab(-0.04(0.02)0.08, tlcolor(gs10) labgap(6-pt) format(%4.2f)) yline(0, lcolor(gs10) lwidth(medthick)) yline(0.017, lcolor(blue) lwidth(medthick)) 
graph export "$out\any_epdsdate.png", replace

use "$work\robust_adm24_epdsweek5",clear
gen w = 5
forvalues i = 6/20 {
append using "$work\robust_adm24_epdsweek`i'"
replace w = `i' if w == .
}
drop if parm == "_cons"
replace t = round(t,0.001)
replace stderr = round(stderr,0.001)
twoway (scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(estimate) mlabp(12) mlabs(vsmall) mlabg(15pt) mlabc(black)) /// 
(scatter estimate w, msize(small) msymbol(Oh) mcolor(red) alwidth(none) lcolor(black) lwidth(thick) mlabel(stderr) mlabp(12) mlabs(vsmall) mlabg(6pt) mlabc(black)) ///
(line estimate w, fcolor(black%10) alwidth(none) lcolor(red) lwidth(thick)) ///
(line max95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)) ///
(line min95 w, fcolor(black%30) alwidth(none) lcolor(gs10) lwidth(medium) lpattern(dash)), ylab(,labcolor(black) glcolor(gs10%33) angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) xlab(5 "5" 10 "10" 15 "15" 20 "20", labcolor(black) glcolor(gs10%33) grid angle(horizontal) tlcolor(gs10) tlwidth(medthick) ticks) graphregion(color(white)) bgcolor(white) ysize(7) xsize(10) yscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) xscale(line noextend nofextend lcolor(gs10) lwidth(medthick)) graphregion(lcolor(white) fcolor(white)) xtitle(,color(black)) ytitle(,color(black)) legend(off) xtitle("Exclude screening scores above X weeks") ytitle("Estimate") ylab(-0.04(0.02)0.06, tlcolor(gs10) labgap(6-pt) format(%4.2f)) yline(0, lcolor(gs10) lwidth(medthick)) yline(0.005, lcolor(blue) lwidth(medthick)) 
graph export "$out\adm_epdsdate.png", replace

